
                                                        MassBay Community College 
ET 111 - iCREAT I 

 

Week 5 LAB:  
Arduino Ultrasonic Sensor 

 

Objectives 
● Learn to use the Ultrasonic sensor 

● Learn to solve problems using the ultrasonic sensor 

Required Resources: 

● Access to the Arduino IDE 

● An Arduino UNO connected to a computer 

● A breadboard, an ultrasonic sensor, jumper wires, a Y 

jumper cable. 

● Access to the iCREAT course in Blackboard 

● Enough time to have fun! 

Part 1. The Ultrasonic Sensor  

This sensor has 4 pins but we will use it as a 3 pin sensor.   

● The echo pin - listens for the returned pulse  

● The trig pin - gets triggered to start the detection 

● 5 v 

● ground 

The ultrasonic range finder detects the distance of objects in front of the sensor for distances 

between 3 cm to up to 4 meters.  It works by sending out a burst of ultrasound to trigger the 

detection, and then listening for the echo when it bounces off an object.  

The Arduino sends a short pulse to the Trig pin to trigger the detection.   It then listens on the 

Echo pin for a pulse.  The duration of the echo pulse is equal to the time taken by the 

ultrasound to travel to the object and back to the sensor.  You can use the speed of sound to 

convert the duration to a distance. 

1. Using a Y jumper cable, connect pin 7 to the Trigger and Echo pins (INPUT/ OUTPUT). 

2. Connect the other pins to power and ground. 

3. In the Arduino IDE under File->Examples->06 Sensors find the ultrasonic sensor 

sketch called PING and load it.  

4. Upload and run the program. 

5. See below to understand the sketch. 

6. Notice the program will display data generated by the sensor to the serial monitor in 

inches and centimeters.  Open it to see the values.  

7. Modify your circuit and introduce an LED with a resistor. 

8. Modify the sketch to control the LED (on) when an object is close to the ultrasonic 

sensor, otherwise turn it off. 

9. Show us your completed program before you leave. 

Understanding the Code! 

const int pingPin = 7;  // This is the pin number of the sensor's output 
 
void setup() { 
   Serial.begin(9600);   // initialize serial communication to display data 
} 
 
void loop() { 
  // establish variables for duration of the ping, and distance result in inches and centimeters 
  long duration, inches, cm; 
 
  // The PING))) is triggered by a HIGH pulse of 2 or more microseconds. 
  // Give a short LOW pulse beforehand to ensure a clean HIGH pulse: 
  pinMode(pingPin, OUTPUT); 
  digitalWrite(pingPin, LOW); 
  delayMicroseconds(2); 
 
  digitalWrite(pingPin, HIGH); 
  delayMicroseconds(5); 
  digitalWrite(pingPin, LOW); 
 
  // The same pin is used to read the signal from the PING))): a HIGH 
  // pulse whose duration is the time (in microseconds) from the sending 
  // of the ping to the reception of its echo off of an object. 
  pinMode(pingPin, INPUT); 
  duration = pulseIn(pingPin, HIGH); 
 
  // use a function to convert the time into a distance 
  inches = microsecondsToInches(duration); 
  cm      = microsecondsToCentimeters(duration); 
 
  Serial.print(inches); 
  Serial.print("in, "); 
  Serial.print(cm); 
  Serial.print("cm"); 
  Serial.println(); 
 
  delay(100); 
} 

long microsecondsToInches(long microseconds) {  // this is a function 
  // According to Parallax's datasheet for the PING))), there are 
  // 73.746 microseconds per inch (i.e. sound travels at 1130 feet per 
  // second).  This gives the distance travelled by the ping, outbound 
  // and return, so we divide by 2 to get the distance of the obstacle. 
  // See: http://www.parallax.com/dl/docs/prod/acc/28015-PING-v1.3.pdf 
  return microseconds / 74 / 2; 
} 
 
long microsecondsToCentimeters(long microseconds) {  // this is a function 
  // The speed of sound is 340 m/s or 29 microseconds per centimeter. 
  // The ping travels out and back, so to find the distance of the 
  // object we take half of the distance travelled. 
  return microseconds / 29 / 2; 
} 

 

How is the Distance Calculated? 

Take a look at the sketch.  The duration value indicates the time it takes the ultrasonic signal 

to travel, bounce off the object and return.  The speed of sound is 340.29 m/s.  What is the 

distance displayed by the sensor? 

Solution: The duration (data read from the echo pin) is in microseconds. The speed of sound 

is 340 meters per second or 29 microseconds per centimeter.  The signal travels to the object 

and back, so the object is actually only half the distance, therefore:   

The distance to the object = (duration / 29) / 2.   

How do we get 29 microseconds per centimeter? 

● Speed of sound is 340.29 meters per second 

● 340.29 x 100 = 34,029 cm per second.  In 1 second sound travels 34,029 cm. 

● How long will it take for sound to travel 1 cm? 

● 1 / 34029 = 0.000029387 seconds for 1 cm.  This is 29.387 microseconds for 1 cm.  

